# $Id$
# Maintainer: Stéphane Gaudreault <stephane@archlinux.org>
# Contributor: Jeff 'codemac' Mickey <jeff@archlinux.org>

pkgname=cppunit
pkgver=1.14.0
pkgrel=1
pkgdesc="A C++ unit testing framework"
arch=('i686' 'x86_64')
url="http://www.freedesktop.org/wiki/Software/cppunit"
license=('LGPL')
depends=('sh' 'gcc-libs')
options=('!emptydirs')
source=(https://dev-www.libreoffice.org/src/${pkgname}-${pkgver}.tar.gz)
sha256sums=('3d569869d27b48860210c758c4f313082103a5e58219a7669b52bfd29d674780')

build() {
   cd ${pkgname}-${pkgver}
   ./configure --prefix=/usr --disable-static
   make
}

check() {
  cd ${pkgname}-${pkgver}
  make check
}

package() {
   cd ${pkgname}-${pkgver}
   make DESTDIR="${pkgdir}" install
}
